

ODS RTF – Erweiterte Möglichkeiten durch direkte RTF Befehle

Katja Glaß
Bayer Schering Pharma AG
Sellerstraße 31
13342 Berlin
Katja.Glass@schering.de

Zusammenfassung

Das Output Delivery System von SAS 9 bietet mit ODS RTF viele Möglichkeiten Microsoft Word-taugliche Dateien zu erstellen. Im Bereich der klinischen Studienauswertung sind die zu erstellenden RTF Dokumente sehr hohen Anforderungen ausgesetzt, die über das Potential von ODS RTF hinausgehen. Mit einigen Hilfsbefehlen von RTF gelingt es in die Zieldokumente viele weitere Extras einzubauen, so dass komplexe und attraktive Berichte mit logischen Tabellen im RTF und somit auch im Word Format herstellbar sind.

Die verschiedenen Layoutmöglichkeiten vom RTF sollen zusammenfassend kurz dargestellt werden. Der Schwerpunkt liegt auf den Besonderheiten von direkten RTF Befehlen, die verwendet werden können um beispielsweise eine Autonummerierung oder andere als die von SAS angebotenen Inhaltsverzeichnisse zu erzeugen. Die Anweisungen, welche dafür notwendig sind, werden vorgestellt. Darüber hinaus wird eine generelle Methode präsentiert, wie neue Befehle gefunden und schlussendlich in die Ausgabe integriert werden können.

Schlüsselwörter: ODS RTF, PROC REPORT

1 Motivation

Das Output Delivery System von SAS in Version 9.1 bietet vielseitige Möglichkeiten Auswertungen auf unterschiedlichste Weise in verschiedenen Formaten darzustellen. Der Hauptvorteil liegt dabei darin, dass logische Tabellen erstellt werden können, die durch Zellen und nicht durch Leerzeichen getrennt sind, welche dann für Formatierungen und Justierungen verwendet werden können.

Für klinische Studien werden umfangreiche Dokumente der Datenauswertung zumeist im Word Format benötigt. Um Layoutelemente, sowie formatierbare und justierbare Tabellen zu erzeugen, die den heutigen Formatanforderungen genügen, bietet sich ODS RTF im Gegensatz zum Listing (ASCII) Format an. Was ODS RTF anbietet, sowie weitere Möglichkeiten, die RTF Ausgabe zu gestalten, soll nun hier erörtert werden.

2 Grundlagen

2.1 Grundoptionen

Zu den wichtigen Grundoptionen für Ausgaben gehören Seiten- und Papiereinstellungen. Im deutschen Raum wird meist A4 und je nach Tabellengrößen Hochformat (portrait) oder Querformat (landscape) verwendet. Diese Optionen sollten einmalig am Anfang einer SAS Sitzung gesetzt werden.

```
OPTIONS PAPERSIZE=A4 ORIENTATION=landscape;
```

2.2 Basis

Um den ODS RTF Ausgabekanal zu öffnen und zu schließen, sind zwei weitere Grundbefehle von Nöten.

```
ODS RTF;  
ODS RTF CLOSE;
```

Alle SAS Ausgaben die zwischen diesen Grundbefehlen erstellt werden, erscheinen in einer RTF Datei. Strukturell ist die Ausgabe dabei identisch mit der Listing Ausgabe. Ist beispielsweise eine einfache Tabelle für einen Datensatzinhalt gewünscht, so kann dies wie folgt realisiert werden:

```
ODS RTF FILE="C:\temp\class.rtf"  
AUTHOR="Katja Glaß"  
TITLE ="Ausgabe von SASHELP.CLASS";  
PROC PRINT DATA=sashelp.class; RUN;  
ODS RTF CLOSE;
```

Der „FILE“ Befehl leitet dabei die Ausgabe in eine bestimmte Datei und nicht in die von SAS angelegte Arbeitsumgebung. Mit „AUTHOR“ und „TITLE“ können Titel und Autor als Metainformationen der Datei mitgegeben werden.

Falls mit allgemeinen TITLES und FOOTNOTES gearbeitet wird, und diese nicht in der Word Kopf- und Fußzeile erscheinen sollen, so ist die Option „BODYTITLE“ zu verwenden. Um jede Tabelle auf einer neuen Seite starten zu lassen, wird „STARTPAGE=YES“ verwendet. Das wertvollste beim ODS ist die Verwendung von Styles, womit generelle Layoutoptionen angewendet werden. Der Style wird bei ODS RTF mit einem STYLE Befehl mitgegeben.

```
ODS RTF FILE="C:\temp\class.rtf"  
BODYTITLE STARTPAGE=YES  
STYLE=journal;  
TITLE1 "Printout of sashelp.class";  
FOOTNOTE1 "Note: Created on Februrary 2007";  
PROC PRINT DATA=sashelp.class; RUN;  
ODS RTF CLOSE;
```

Grundlagen zu den ODS RTF Styles finden sich in diversen Dokumenten, zum Beispiel im SUGI Paper 125-29 [1].

2.3 ODS Escapechar

Um neben den ODS RTF Elementen sehr flexibel sein zu können, ist es möglich, in ODS RTF direkte Feldfunktionen sowie Style und RTF Befehle einzubauen. Hierzu muss zunächst ein Zeichen definiert werden, das durch SAS als Sonderzeichen interpretiert wird, um den folgenden Code besonders zu behandeln. Die Definition eines solchen Zeichens findet über die ESCAPECHAR Option statt:

```
ODS ESCAPECHAR='^';
```

In allen Ausgaben sind nun über dieses Sonderzeichen verschiedene Elemente einbindbar. (Beispiel in 3.2)

3 Styling mit ODS RTF

3.1 Allgemeine Layoutoptionen

Für Titel und Fußnoten bietet SAS einige Standardlayouteinstellungen. So können zum Beispiel Fettschrift (bold) aktiviert werden, Farben (color) und Hintergrundfarben (bcolor) gesetzt werden, die Schriftgröße (font) sowie die Höhe (height) angegeben werden. Ebenso werden verschiedene Ausrichtungen (justify), Unterstreichungen (underlin) und das Verlinken (link) angeboten. [2]

```
ODS RTF FILE="C:\temp\example.rtf";
TITLE1 BOLD COLOR=blue BCOLOR=yellow "Dicker, blauer Titel auf gelbem
                                     Hintergrund";
TITLE2 FONT=arial HEIGHT=20 UNDERLIN=1 "Großer unterstrichener arial Titel";
TITLE3 JUSTIFY=R LINK="www.sas.com" "Rechtsbündiger SAS-Link Titel";
PROC PRINT DATA=sashelp.class;RUN;
ODS RTF CLOSE;
```

Standardlayoutoptionen für Tabellen werden ebenfalls von SAS angeboten. Da die meisten Ausgaben mit PROC REPORT realisiert werden, gehe ich nur auf diese Prozedur näher ein. Die Layoutoptionen können für verschiedene Bereiche des Reports verwendet werden. Die geläufigsten sind der Kopfzeilenbereich (header), der Spaltenbereich (column), der Linienbereich (lines) und der komplette Reportbereich (report). Hierzu können vielseitige Optionen wie Farben, Schriftarten und -größen, Ausrichtungen, Zellenabstände und vieles mehr eingestellt werden. Eine komplette Auflistung ist in der SAS Dokumentation zu finden [3].

Nachfolgend ein Beispiel, um die vielfältigen Möglichkeiten anzudeuten:

```
ODS RTF FILE="C:\temp\example.rtf";
PROC REPORT DATA=sashelp.shoes NOWINDOWS HEADLINE MISSING
```

```

style(header)=[ just=center font_size=10pt
                font_weight=bold background=white]
style(report)=[cellpadding=6pt];
column region subsidiary product stores sales inventory returns;
DEFINE region      / display;
DEFINE stores      / display;
DEFINE inventory  / display STYLE={FONT_STYLE=italic};
DEFINE returns    / display STYLE={CELLWIDTH=3cm};
compute region;
    CALL DEFINE(_col_, "Style", "STYLE=[indent=1cm]");
endcomp;
compute stores;
    IF stores < 4
    THEN DO;
        CALL DEFINE(1, "Style", "STYLE=[background=yellow]");
    END;
endcomp;
RUN;
ODS RTF CLOSE;

```

3.2 Eingebettete Layoutoptionen

Die vorgestellten Style-Elemente (siehe 3.1) können auch an verschiedenen anderen Stellen innerhalb von Textbausteinen über des Escapechar und der Kennung „S“ verwendet werden. So lassen diese sich wie folgt in „Title“ und „Footnote“ Zeilen einbauen:

```

ODS RTF FILE = "C:\temp\blub.rtf" bodytitle;
title1 "Standard Titel";
title2 "^S={indent=.12in} Eingerückter Titel";
title3 "^S={just=right} Rechtsbündiger Titel";
title4 "^S={font_size=40pt} Großer Titel";
title5 "^S={font=('ARIAL',20pt,bold)} Großer, dicker und ari-
ler Titel";
footnote1 "^S={preimage='H:\My Documents\My Pictures\fische-
7.jpg'} Das Bild";
PROC PRINT DATA=sashelp.class;RUN;
ODS RTF CLOSE;

```

Analog können diese Elemente durch ODS RTF TEXT verwendet werden:

```

ODS RTF FILE="C:\temp\blub.rtf" bodytitle;
ODS RTF TEXT = "Standard Text";
ODS RTF TEXT = "^S={indent=.12in} Eingerückter Text";
ODS RTF TEXT = "^S={just=right} Rechtsbündiger Text";
ODS RTF TEXT = "^S={font_size=20pt} Großer Text";

```

```
ODS RTF TEXT = "^S={font=('ARIAL',20pt,bold)} Großer, dicker
und arialer Text";
ODS RTF CLOSE;
```

In den eigentlichen Ausgaben, beispielsweise bei Proc Report, können die Elemente ebenfalls genutzt werden:

```
ODS RTF FILE="C:\temp\example.rtf";
PROC REPORT DATA=sashelp.class NOWINDOWS HEADLINE MISSING;
  COLUMN sex name age height weight;
  DEFINE sex / group noprint;
  COMPUTE BEFORE sex;
    LINE '^S={just=left font_weight=bold}Sex=' sex $1.;
  ENDCOMP;
RUN;
ODS RTF CLOSE;
```

Dabei müssen die Textbausteine nicht unbedingt immer am Anfang stehen. So können auch nur einzelne Buchstaben fett und der Rest wieder normal formatiert werden:

```
ODS RTF FILE="C:\temp\example.rtf";
PROC REPORT DATA=sashelp.class NOWINDOWS HEADLINE MISSING;
  COLUMN sex name age height weight;
  DEFINE sex / group noprint;
  COMPUTE BEFORE sex;
    LINE '^S={just=left
font_weight=bold}S^S={font_weight=medium}ex=' sex $1.;
  ENDCOMP;
RUN;
ODS RTF CLOSE;
```

Manchmal ist die Reihenfolge der Befehle von entscheidender Bedeutung. So ist die Ausrichtung im folgenden Beispiel nicht linksbündig wie angenommen, obwohl nur ein zusätzlicher LINE Befehl eingefügt wurde. Wenn eine Linksbündigkeit gewünscht ist, so muss dies im ersten LINE Statement berücksichtigt werden.

```
ODS RTF FILE="C:\temp\example.rtf";
PROC REPORT DATA=sashelp.class NOWINDOWS HEADLINE MISSING;
  column sex name age height weight;
  define sex / group noprint;
  compute before sex;
    line '';
    line '^S={just=left font_weight=bold}Sex=' sex $1.;
  endcomp;
RUN;
ODS RTF CLOSE;
```

4 Styling mit RTF

4.1 RTF Tags einbetten

Die interessanteste und flexibelste Möglichkeit das Dokument zu gestalten, bietet SAS durch die Möglichkeit der Nutzung von direkten RTF Befehlen und Feldfunktionen. Verschiedene RTF Befehle (Tags) können auf unterschiedliche Weise in das Dokument eingebettet werden. SAS maskiert dabei grundsätzlich die Sonderzeichen, die ein Tag auslösen würde. Dies kann in einem Style durch die „protectspecialcharacter“ Option ausgeschaltet werden. Um Fehler zu vermeiden, sollte SAS jedoch nur mitgeteilt werden, dass an entsprechenden Stellen die Tags nicht maskiert werden sollen:

Wenn eine Textausgabe mit “{“ startet und “}“ endet, so erkennt SAS, dass der Inhalt nicht maskiert werden soll und die beinhaltenden Tags werden vom RTF interpretiert. Alternativ können die Befehle durch den ESCAPECHAR sowie dem folgenden Befehlsword R/RTF von der Maskierung ausgenommen werden.

Der Tag „\b“ bewirkt, dass der folgende Text fett geschrieben wird. Im folgenden Beispiel wird im ersten Fall \b maskiert und erscheint als Text ohne jegliche Auswirkung. In Fall 2 und 3 erscheint das letzte Wort in Fettschrift.

- 1) ODS RTF TEXT = "Das dritte \b Wort";
- 2) ODS RTF TEXT = "{Das dritte \b Wort}";
- 3) ODS RTF TEXT = "Das dritte ^R/RTF'\b 'Wort";

Übliche RTF Tags werden von SAS bei ODS FAQ and Concepts [2] erwähnt. Eine ausführliche Dokumentation über RTF findet sich in der RTF Spezifikation [4]. Sehr hilfreich bei der Suche nach bestimmten Tags sind Programme, deren Dokumente im RTF Format gespeichert werden. Hierzu gehören unter anderem Microsoft Word und Open Office. Das RTF Dokument kann dabei analysiert werden, um zu ermitteln, welche Tags an welcher Position und in welcher Reihenfolge benutzt werden müssen um ein bestimmtes Ergebnis zu erzielen. Leider verwenden die Programme wesentlich mehr Tags als eigentlich notwendig, so dass oft viel suchen und ausprobieren notwendig ist.

4.2 Feldfunktionen

Sehr nützliche Tags sind die RTF Feldfunktionen. Sämtliche vom öffnenden Programm angebotenen Funktionen lassen sich in das Dokument einbauen. Sehr beliebte Felder sind unter anderem {page} für die aktuelle Seite, {pageof} für die Anzahl der Seiten, aber auch {filename} als Dateiname und {createdate} / {savedate} als nützliche Information, die dargestellt werden sollen.

Bei Microsoft Word können die Feldnamen einfach ermittelt werden, indem ein solches Feld in ein beliebiges Dokument eingefügt wird (Einfügen -> Feld ... -> entsprechendes Feld wählen) und dann mit Alt+F9 der Feldname angezeigt wird. Soll zum Beispiel ein komplexes Inhaltsverzeichnis mit unterschiedlichen Styles verwendet werden, so kann die zugehörige Feldfunktion im RTF Dokument einfach verwendet werden.

Diese Feldspezifikationen müssen dann in RTF Befehlsword eingebettet werden, die implizieren, dass es sich um Felder handelt. Die zugehörigen Tags lauten

„`{\field{\fldinst <FELDNAME> }}`“. Um die Feldfunktionstags im RTF Dokument einzubetten, wird wieder eine der zwei vorgestellten Optionen verwendet. Folgendes Beispiel demonstriert die Nutzung von Feldfunktionen:

```

OPTIONS NODATE NONUMBER;
ODS RTF FILE="C:\temp\example.rtf";
TITLE1 j=r "{Seite {\field{\fldinst{page}}} von
{\field{\fldinst{numpages}}}}";
FOOTNOTE "Datei ^R/RTF' {\field{\fldinst{filename}}}' erstellt
am ^R/RTF' {\field{\fldinst{createdate}}}' gespeichert am
^R/RTF' {\field{\fldinst{savedate \@ "d.M.yyyy"}}}'";

PROC REPORT DATA=sashelp.class NOWINDOWS HEADLINE MISSING;
    column sex name age height weight;
    compute before _page_;
        line 'Table ^R/RTF' {\field{\fldinst{SEQ table}}}' :
Auflistung der Klasse';
    endcomp;
RUN;

ODS RTF CLOSE;

```

Der Titel beinhaltet „Seite X von Y“. In der Fußnote sind Informationen über Dateiname, Erstelldatum und Speicherdatum eingebettet. Zu beachten ist hierbei, dass Formatierungen vorgenommen werden können. So besteht das Speicherdatum nur aus Tag Monat Jahr und nicht zusätzlich aus der Uhrzeit. In der Tabelle befindet sich eine automatische Zählvariable. Diese kann für Autonummerierungen verwendet werden.

4.3 Inhaltsverzeichnisse

SAS bietet experimentell ein einfaches Inhaltsverzeichnis durch „CONTENTS“ an [6]. In einfachen Fällen bietet dieses ein brauchbares Resultat, für alle anderen Fälle ist jedoch eigenes Erstellen unverzichtbar.

Ein Inhaltsverzeichnis entspricht in RTF einer Feldfunktion. Das Standardinhaltsverzeichnis von Microsoft Word besteht aus der Feldfunktion `{TOC \o "1-3" \h \z \u}`. Alternativ können die Namen der Word Styles sowie die gewünschte Leveltiefe selbst angegeben werden `{TOC \t "Überschrift 1;1;Überschrift 2;2;Überschrift 3;3"}`. Diese Art Inhaltsverzeichnisse zu erstellen kann auch dazu verwendet werden, um ein Tabellenverzeichnis zu erstellen. Sofern die Tabellenüberschriften einen entsprechenden Word Style besitzen, wie beispielsweise „table“ im Beispiel, kann ein solches Verzeichnis durch `{ TOC \h \z \c "Table" }` erstellt werden.

Die Anwendung von Word Styles in ODS RTF Dokumente kann in verschiedenen Artikeln nachgelesen werden, siehe hierzu [7] und [8].

Das Inhaltsverzeichnis wird dann in die RTF Feld Befehle eingebettet. Da „\“ vom RTF als Sonderzeichen interpretiert wird, muss dieses Zeichen für das RTF Dokument zusätzlich maskiert werden.

Außerdem kann bei Feldern ein Text angegeben werden, der bis zur Aktualisierung erscheint. Im folgendem Beispiel soll „Inhaltsverzeichnis“ das Feld beschreiben.

```
ODS RTF FILE="C:\temp\example.rtf" bodytitle;
ODS RTF TEXT='^R/RTF"{\field{\fldinst{TOC \o ''1-3'' \h \z
\u}}{\fldrslt {Inhaltsverzeichnis }}}}'";
ODS RTF CLOSE;
```

Nachteilig am ODS RTF TEXT ist die Verwendung einer Zelle, in der das Inhaltsverzeichnis erstellt wird. Da diese Zelle viel zu schmal für das ausgefüllte Verzeichnis ist, bietet es sich an, diese Zelle auf 100% der Seitengröße anzupassen. Hierzu kann das Styleelement „OUTPUTWIDTH“ verwendet werden:

```
ODS RTF FILE="C:\temp\example.rtf" bodytitle;
ODS RTF TEXT='^S={outputwidth=100%
just=1}^R/RTF"{\field{\fldinst{TOC \o ''1-3'' \h \z
\u}}{\fldrslt {Inhaltsverzeichnis }}}}'";
ODS RTF CLOSE;
```

4.4 Linien Tags

Wenn ein bestimmtes Layout mit bestimmter Linienführung gewünscht ist, so kann dies nicht durch die ODS Styles erreicht werden. Es kann notwendig werden manuell zu sagen, dass an dieser oder jener Stelle eine Linie sein soll. Leider kann die Linienführung der Tabellenzellen in SAS nicht direkt eingestellt werden. Die Alternative Paragraphenlinien zu verwenden, ist dagegen erfolgreich. Hierzu existiert ein interessantes Paper von Steve Prust [5]. Folgendes Beispiel stellt die Funktionsweise und Tags kurz dar:

```
ODS RTF FILE="C:\temp\example.rtf" style=journal;
PROC REPORT DATA=sashelp.class NOWINDOWS HEADLINE MISSING;
  column sex name age height weight;
  define sex / group;
  compute before _page_;
    line '^R/RTF"\brdrb\brdrs\brdrw15 "Table: Auflistung';
  endcomp;
  compute after sex;
    line '^R/RTF"\brdrb\brdrs\brdrw15"';
  endcomp;
RUN;
ODS RTF CLOSE;
```

In dem Beispiel wird eine Linie unterhalb des Tabellentitels und über jede Trennung des Geschlechts gezogen. Wenn im Bearbeitungsprogramm einzelne Linien nicht dargestellt werden, liegt das zumeist daran, dass aktivierte Gitternetzlinien die eigentlichen Linien überblenden. In der Seitenvorschau oder beim Ausblenden der Gitternetzlinien sind diese Linien dann wieder sichtbar.

4.5 Zeileneinzüge im RTF

Eine weitere oft gewünschte Funktion sind bedingte Einrückungen. Im ASCII wurden führende Leerzeichen zum Einrücken verwendet. Zwar kann die Option ASIS verwendet werden, um führende Leerzeichen nicht zu löschen, oft sind jedoch statt führenden Leerzeichen Zeileneinzüge gewünscht. Das ODS RTF Styleelement „INDENT“ kann für Einrückungen verwendet werden, ist jedoch oft nicht flexibel genug, um beispielsweise je nach Anzahl der führenden Leerzeichen einen Zeileneinzug zu erstellen. Der RTF Tag für Zeileneinzüge ist „li“ plus eine Zahl, welche die Breite des Zeileneinzugs ausdrückt.

Um ein Beispiel zu zeigen, wird zunächst folgender Beispieldatensatz erstellt:

```
PROC SORT DATA=sashelp.class OUT=class; BY sex name; RUN;
DATA class2;
  SET class;
  BY sex;
  LENGTH text $20;
  text = '    ' || TRIM(name);
  sort=2;
  OUTPUT;
  IF first.sex
  THEN DO;
    IF sex = "F" THEN text = "Female";
    ELSE text = "Male";

    sort = 1;
    age=.; height=.; weight=.;
    OUTPUT;
  END;
RUN;
PROC SORT DATA=class2; BY sex sort; RUN;
OPTIONS MISSING=' ';
```

Die „ASIS“ Option ist im folgendem Beispiel dargestellt:

```
ODS RTF FILE="C:\temp\example.rtf" style=journal;
PROC REPORT DATA=class2 NOWINDOWS HEADLINE MISSING
  style(column)={asis=on};
  column text age height weight;
RUN;
ODS RTF CLOSE;
```

Um Zeileneinzüge abhängig von den führenden Leerzeichen zu erstellen, kann folgender Code verwendet werden:

```
ODS RTF FILE="C:\temp\example.rtf" style=journal;
PROC REPORT DATA=class2 NOWINDOWS HEADLINE MISSING;
  COLUMN text age height weight;
```

```
DEFINE sex / GROUP noprint;
COMPUTE text;
    spacediff = rxmatch(rxparse("^' '"), text) * 70;
    IF spacediff > 70
        THEN CALL DEFINE(_col_, "Style",
            "STYLE=[cellwidth=3cm
            pretext='^R/RTF""\keepn\li" ||
            compress(spacediff) || " "'');
ENDCOMP;
RUN;
ODS RTF CLOSE;
```

Zunächst wird über einen regulären Ausdruck ermittelt, wie viele führende Leerzeichen die Spalte besitzt. Diese wird dann mit einer Konstanten multipliziert. Die 70 wurde in diesem Fall gewählt, da sie in etwa der Breite eines Leerzeichens (Times New Roman Schriftgröße 10) entspricht. Sofern führende Leerzeichen vorhanden sind, wird die aktuelle Zelle eingerückt, um den entsprechend ausgerechneten Wert (pretext). Da SAS bei der Spaltenberechnung Zeileneinrückungen nicht beachtet – weder wenn das „INDENT“ Styleelement noch wenn der RTF Befehl verwendet wird – muss eine Spaltenbreitenvorgabe meist manuell angegeben werden. Damit können unerwünschte Zeilenumbrüche verhindert werden.

5 Postprocessing

Die letzte Möglichkeit, das erstellte Dokument zu optimieren bietet das Postprocessing. Alles was noch nicht umgesetzt werden konnte, muss nun hier nachbearbeitet werden.

So gibt es bei der Autonummerierung beispielsweise das Problem, dass wenn eine Tabelle in mehrere logisch geteilt wird, die Überschrift und somit auch das Tabellenummerierungsfeld wiederholt wird. In der Nachbearbeitung können entweder wiederholte Überschriften entfernt oder die Zählvariable durch eine Referenz auf die vorherige Nummer erstellt werden.

SAS erstellt auch bei der „BODYTITLE“ Option für jeden neuen Tabellenabschnitt eine eigene Sektion mit eigener Kopf- und Fußzeile. Diese leeren definierten Bereiche können ebenso entfernt werden.

Ein weiterer Punkt ist der Seitenumbruch. Oft müssen Seitenumbrüche entfernt oder hinzugefügt werden. Auch hierzu bietet sich Postprocessing an. Leider funktioniert die ODS RTF STARTPAGE=NOW nicht in jedem Fall und bei zu breiten Tabellen wird der Seitenumbruch zwischen den geteilten Bereichen oft vermisst.

Ob für die Nachverarbeitung ein DATA STEP, eine andere Programmiersprache wie PEARL oder die Microsoft Word integrierte VBA Sprache verwendet wird, ist dabei dem Programmierer überlassen. Hierbei zählen vor allem Performance, Vertrautheit der Nutzung und die Sicherheit der Ergebnisse.

6 Fazit und Ausblick

Mit den hier vorgestellten Methoden lassen sich mit Hilfe von ODS und eigenen RTF Befehlen und dem Postprocessing alles realisieren, was RTF zu bieten hat. Das ODS System von SAS bietet dabei sehr viele Grundoptionen. Im Zuge der Erweiterungen sind wir gespannt, was für weitere Neuerungen noch eingebaut werden, so dass am Ende weniger manuelle Nacharbeiten in Form von Postprocessing von Nöten sind.

Das TAGSET RTF von ODS MARKUP könnte eine wesentliche Unterstützung sein notwendige RTF Befehle im Tagset aufzunehmen, statt in dem Report einzubetten. Wir sind gespannt, wie flexibel diese Möglichkeit mit dem produktiven Release in SAS 9.2 sein wird. [9]

Zusammenfassend kann festgestellt werden, dass SAS verschiedene Komplexitätsstufen anbietet um RTF zu erstellen, die unterschiedlich flexibel sind. Zunächst sollte versucht werden alles durch reines ODS RTF zu realisieren. Der Quellcode ist leicht verständlich und nachvollziehbar. Im zweiten Schritt sollte versucht werden die fehlenden Funktionalitäten mit direkten RTF Befehlen abzubilden. Misslingt auch dies, so sollte als letzte Maßnahme das Postprocessing verwendet werden, das schnell sehr komplex wird. Mit diesen drei Möglichkeiten kann flexibel alles erreicht werden, was RTF zu bieten hat.

Literatur

- [1] Lauren Haworth [2004] „SAS with Style: Creating your own ODS Style Template for RTF Output“, SUGI 29 – Paper 125-29,
<http://www2.sas.com/proceedings/sugi29/125-29.pdf>
- [2] SAS Support – “ODS FAQ and Concepts – The RTF Destination” -
http://support.sas.com/rnd/base/topics/templateFAQ/Template_rtf.html
- [3] SAS 9.3.1 Onlinedokumentation, “Concepts: REPORT Procedure” -
<http://support.sas.com/onlinedoc/913/getDoc/de/proc.hlp/a000146851.htm>
- [4] Rich Text Format (RTF) Specification, version 1.6, Microsoft MSDN -
[http://msdn2.microsoft.com/en-us/library/aa140277\(office.10\).aspx](http://msdn2.microsoft.com/en-us/library/aa140277(office.10).aspx)
- [5] Steve Prust [2005] “Experiences of using ODS : moving from ASCII to RTF output“, PHUSE 2005 – Paper TS06 -
<http://www.lexjansen.com/phuse/2005/ts/ts06.pdf>
- [6] SAS Support – “Experimental RTF Features in SAS 9.1” -
<http://support.sas.com/rnd/base/topics/odsrtf/rtf901.html>
- [7] Lauren Haworth [2005] „Applying Microsoft Word Styles to ODS RTF Output“, SUGI 30 – Paper 043-30 -
<http://www2.sas.com/proceedings/sugi30/043-30.pdf>
- [8] Elizabeth Axelrod, David Shamlin [2004] „Skinning the Cat This Way and That: Using ODS to Create Word Documents That Work for You“, SUGI 29 – Paper 084-29 -
<http://www2.sas.com/proceedings/sugi29/084-29.pdf>
- [9] Wayne Hester [2006] “Teaching Your RTF Tagset to Do Clever Tricks” – SUGI 31 – Paper 067 – 31 -
<http://www2.sas.com/proceedings/sugi31/067-31.pdf>